



TSwap Potocol Audit Report

Version 1.0

What Fate

August 22, 2025

Protocol Audit Report

What Fate

August 22, 2025

Lead Auditors: - What Fate

Table of Contents

- Table of Contents
- Protocol Summary
- Disclaimer
- Risk Classification
- Audit Details
 - Scope
 - Roles
- Executive Summary
 - Issues found
- Findings
- High Severity
 - [H-1] Protocol invariant broken due to bonus token giveaway in `TSwapPool::_swap`
 - [H-2] Miscalculated fee in `TSwapPool::getInputAmountBasedOnOutput` leads to excessive user charges
 - [H-3] Lack of slippage protection in `TSwapPool::swapExactOutput` causes users to potentially receive way fewer tokens
 - [H-4] Incorrect function call in `TSwapPool::sellPoolTokens` causes mismatched swap direction and wrong token amounts

- Medium Severity
 - [M-1] `TSwapPool::deposit` does not enforce `deadline`, risking post-deadline execution
- Low Severity
 - [L-1] Liquidity token symbol uses `.name()` instead of `.symbol()` in `PoolFactory::createPool`
 - [L-2] `TSwapPool::LiquidityAdded` event has parameters out of order
 - [L-3] `TSwapPool::swapExactInput` returns default zero due to uninitialized output variable
- Informational
 - [I-1] `PoolFactory::PoolFactory__PoolDoesNotExist` is unused
 - [I-2] Missing zero-address validation in `PoolFactory` constructor
 - [I-3] Missing zero-address validation in `TSwapPool` constructor
 - [I-4] Event `TSwapPool::Swap` has more than 3 parameters and should index key fields
 - [I-5] Missing `deadline` NatSpec in `TSwapPool::swapExactOutput`
 - [I-6] CEI pattern not followed in `TSwapPool::deposit`
 - [I-7] Missing NatSpec in `TSwapPool` functions
- Gas Optimization
 - [G-1] `TSwapPool::swapExactInput` can be `external`

Protocol Summary

This project is meant to be a permissionless way for users to swap assets between each other at a fair price. You can think of T-Swap as a decentralized asset/token exchange (DEX). T-Swap is known as an Automated Market Maker (AMM) because it doesn't use a normal "order book" style exchange, instead it uses "Pools" of an asset. It is similar to Uniswap. To understand Uniswap, please watch this video: [Uniswap Explained](#)

Disclaimer

The Hash Horizon team makes all effort to find as many vulnerabilities in the code in the given time period, but holds no responsibilities for the findings provided in this document. A security audit by the team is not an endorsement of the underlying business or product. The audit was time-boxed and the review of the code was solely on the security aspects of the Solidity implementation of the contracts.

Risk Classification

		Impact		
		High	Medium	Low
Likelihood	High	H	H/M	M
	Medium	H/M	M	M/L
	Low	M	M/L	L

We use the CodeHawks severity matrix to determine severity. See the documentation for more details.

Audit Details

- Commit Hash: 55d1e086ed0917fd055b14f63099c2342eb6b86a

Scope

```
1 ./src/  
2 #-- PoolFactory.sol  
3 #-- TSwapPool.sol
```

Roles

- Liquidity Providers: Users who have liquidity deposited into the pools. Their shares are represented by the LP ERC20 tokens. They gain a 0.3% fee every time a swap is made.
- Users: Users who want to swap tokens.

Executive Summary

The audit of `TSwapPool` and `PoolFactory` revealed critical vulnerabilities affecting fund safety and protocol integrity. Bonus token distributions can break the AMM invariant, allowing attackers to drain liquidity, while fee miscalculations and missing slippage protections expose users to financial loss. Other issues include incorrect function calls, unused parameters, and missing documentation,

which reduce usability and maintainability. Addressing these findings is essential before mainnet deployment to ensure security, predictable behavior, and user trust.

Issues found

Severity	Number of issues found
High	4
Medium	1
Low	3
Info	7
Gas	1
Total	16

Findings

High Severity

[H-1] Protocol invariant broken due to bonus token giveaway in TSwapPool::_swap

Description: Giving bonus tokens from the pool breaks the invariant $x * y = k$, because the bonus tokens distributed to the user reduce the pool reserves outside of the swap formula, leading to a mismatch between actual balances and the expected protocol state. This undermines the AMM pricing mechanism, as the pool can no longer guarantee that its reserves reflect the invariant relationship.

```
1      swap_count++;
2      if (swap_count >= SWAP_COUNT_MAX) {
3          swap_count = 0;
4          outputToken.safeTransfer(msg.sender, 1
5                                   _000_000_000_000_000_000);
6      }
```

Impact: A malicious user can repeatedly perform swaps with minimal input amounts. Every 10th swap grants the attacker bonus tokens directly from the pool reserves. Over time, this mechanism enables the attacker to drain the pool completely, causing liquidity providers to lose their deposited funds. The attack is highly cost-efficient, since the attacker only needs to cover the swap fees, while the bonus tokens received far exceed the amounts spent.

Proof of Concept:

1. Liquidity provider deposits equal amounts of poolToken and WETH into the pool using `TSwapPool::deposit`.
2. Malicious user performs repeated swaps, triggering the bonus token mechanism. Every 10th swap grants bonus tokens directly from the pool reserves.
3. The repeated bonus payouts allow the attacker to drain the pool, violating the $x * y = k$ invariant.
4. Post-conditions:
 - User balance increased beyond normal swap gains.
 - Pool reserves decreased below expected invariant.

Proof of Code

Code

```
1      function test_swapExactInput_InvariantViolationByBonus() public {
2          uint256 SWAP_AMOUNT = 1e17;
3          uint256 TOKEN_AMOUNT = 10e18;
4
5          // --- Step 1: Deposit liquidity ---
6          // Liquidity provider deposits equal amounts of poolToken and
7          // WETH into the pool
8          // This sets up the initial pool reserves for swaps
9          vm.startPrank(LiquidityProvider);
10         weth.approve(address(pool), TOKEN_AMOUNT);
11         poolToken.approve(address(pool), TOKEN_AMOUNT);
12         pool.deposit(TOKEN_AMOUNT, TOKEN_AMOUNT, TOKEN_AMOUNT, uint64(
13             block.timestamp));
14         vm.stopPrank();
15
16         // --- Step 2: Perform repeated swaps to trigger bonus tokens ---
17         // The attacker executes many swaps in order to trigger the
18         // bonus token mechanism
19         // Every 10th swap grants bonus tokens directly from the pool
20         // reserves
21         vm.startPrank(user);
22
23         uint256 startingUserPoolTokenBalance = poolToken.balanceOf(user);
24         uint256 startingUserWethBalance = weth.balanceOf(user);
25
26         uint256 startingPoolPoolTokenBalance = poolToken.balanceOf(
27             address(pool));
28         uint256 startingPoolWethBalance = weth.balanceOf(address(pool));
29     }
```

```
25      // Approve large amounts for repeated swapping
26      weth.approve(address(pool), 1000e18);
27      poolToken.approve(address(pool), 1000e18);
28
29      // Execute multiple swaps to trigger bonus payouts repeatedly
30      for (uint256 i = 0; i < 60; i++) {
31          pool.swapExactInput(poolToken, SWAP_AMOUNT, weth, 0, uint64
32              (block.timestamp));
33          pool.swapExactInput(weth, SWAP_AMOUNT, poolToken, 0, uint64
34              (block.timestamp));
35      }
36      // Additional swaps to ensure bonus mechanism triggers more
37      // times
38      for (uint256 i = 0; i < 10; i++) {
39          pool.swapExactInput(poolToken, SWAP_AMOUNT, weth, 0, uint64
40              (block.timestamp));
41      }
42      vm.stopPrank();
43
44      uint256 endingUserPoolTokenBalance = poolToken.balanceOf(user);
45      uint256 endingUserWethBalance = weth.balanceOf(user);
46
47      uint256 endingPoolPoolTokenBalance = poolToken.balanceOf(
48          address(pool));
49      uint256 endingPoolWethBalance = weth.balanceOf(address(pool));
50
51      // --- Step 3: Verify pool invariant is violated ---
52
53      // User received extra pool tokens due to the bonus mechanism
54      assertLt(startingUserPoolTokenBalance,
55          endingUserPoolTokenBalance);
56      // User received extra WETH due to the bonus mechanism
57      assertLt(startingUserWethBalance, endingUserWethBalance);
58      // Pool's poolToken reserves decreased beyond what the
59      // invariant allows
60      assertGt(startingPoolPoolTokenBalance,
61          endingPoolPoolTokenBalance);
62      // Pool's WETH reserves decreased beyond what the invariant
63      // allows
64      assertGt(startingPoolWethBalance, endingPoolWethBalance);
65  }
```

Recommended Mitigation: Do not distribute bonus tokens directly from the pool, as this breaks the $x * y = k$ invariant. Instead, issue bonus tokens from a separate reward reserve contract. Implement strict limits on both the frequency and the amount of bonus distributions. Bonuses should be rare and pre-determined, preventing an attacker from draining the pool through repeated swaps. Additionally, ensure that all bonus mechanisms are tested against the pool invariant to avoid unintended depletion of reserves.

[H-2] Miscalculated fee in TSwapPool::getInputAmountBasedOnOutput leads to excessive user charges

Description: The `getInputAmountBasedOnOutput` is incorrectly scales fees by 10_000 instead of 1_000. This overestimation causes the protocol to require more input tokens than intended, resulting in users being overcharged and losing funds through inflated fees.

As a result, users are forced to deposit more tokens than necessary, effectively being overcharged by the protocol.

```
1     function getInputAmountBasedOnOutput(  
2         uint256 outputAmount,  
3         uint256 inputReserves,  
4         uint256 outputReserves  
5     )  
6     public  
7     pure  
8     revertIfZero(outputAmount)  
9     revertIfZero(outputReserves)  
10    returns (uint256 inputAmount)  
11    {  
12 -->    return ((inputReserves * outputAmount) * 10000) / ((  
        outputReserves - outputAmount) * 997);  
13    }
```

Impact:

- Users consistently lose funds due to inflated swap costs.
- Protocol collects unintended excess fees, creating accounting mismatches and trust issues.

Proof of Concept:

The following test demonstrates the discrepancy between the correct formula and the current implementation:

Proof of Code

Code

```
1     function test_getInputAmountBasedOnOutput_feeMiscalculation()  
2     public {  
3         uint256 initialLiquidity = 10e18;  
4         uint256 desiredOutputAmount = 2e18;  
5  
6         uint256 feeDenominator = 1000;  
7         uint256 feePercent = 3; // 0.3%  
8         uint256 feeMultiplier = feeDenominator - feePercent;  
9  
        // --- Arrange: provide initial liquidity to the pool ---
```



```
10     vm.startPrank(liquidityProvider);
11     weth.approve(address(pool), initialLiquidity);
12     poolToken.approve(address(pool), initialLiquidity);
13     pool.deposit(initialLiquidity, initialLiquidity,
14                 initialLiquidity, uint64(block.timestamp));
15     vm.stopPrank();
16
17     uint256 inputReserves = weth.balanceOf(address(pool));
18     uint256 outputReserves = poolToken.balanceOf(address(pool));
19
20     // Expected formula with correct fee scaling (1000 basis points
21     // , fee 0.3%)
22     uint256 expectedInputAmount = ((inputReserves *
23                                     desiredOutputAmount) * feeDenominator)
24                                     / ((outputReserves - desiredOutputAmount) * feeMultiplier);
25
26     // Act: call the vulnerable function
27     uint256 actualInputAmount = pool.getInputAmountBasedOnOutput(
28         desiredOutputAmount, inputReserves, outputReserves);
29
30     // --- Assert: the function takes MORE than it should due to
31     // miscalculation ---
32     assert(actualInputAmount != expectedInputAmount);
33 }
```

Recommended Mitigation: Ensure fees are applied using the correct basis point denominator (1,000), consistent with Uniswap-like models.

```
1 - return ((inputReserves * outputAmount) * 10_000) / ((outputReserves
2 -   - outputAmount) * 997);
3 + return ((inputReserves * outputAmount) * 1_000) / ((outputReserves
4 -   - outputAmount) * 997);
```

[H-3] Lack of slippage protection in `TSwapPool::swapExactOutput` causes users to potentially receive way fewer tokens

Description: The `swapExactOutput` function does not include any sort of slippage protection. This function is similar to what is done in `TSwapPool::swapExactInput`, where the function specifies a `minOutputAmount`, the `swapExactOutput` function should specify a `maxInputAmount`.

Impact: If market conditions change before the transaction precesses, the user could get a much worse swap

Proof of Concept:

1. The price of 1 WETH right now is 1,000 USDC
2. User inputs a `swapExactOutput` looking for 1 WETH

- inputToken = USDC
 - outputToken = WETH
 - outputAmount = 1
 - deadline = whatever
3. The function does not offer a maxInput amount
 4. As the transaction is pending in the mempool, the market changes! And the price moves HUGE
-> 1 WETH is now 10,000 USDC. 10x more than the user expected
 5. The transaction completes, but the user sent the protocol 10,000 USDC instead of the expected 1,000 USDC

Recommended Mitigation: We should include a `maxInputAmount` so the user only has to spend up to a specific amount, and can predict how much they will spend on the protocol.

```
1     function swapExactOutput(  
2         IERC20 inputToken,  
3     +     uint256 maxInputAmount,  
4     .  
5     .  
6     .  
7         inputAmount = getInputAmountBasedOnOutput(outputAmount,  
8             inputReserves, outputReserves);  
8     +     if(inputAmount > maxInputAmount){  
9     +         revert();  
10    +     }  
11    _swap(inputToken, inputAmount, outputToken, outputAmount);
```

[H-4] Incorrect function call in `TSwapPool::sellPoolTokens` causes mismatched swap direction and wrong token amounts

Description: The `sellPoolTokens` function is designed to let users redeem their pool tokens (`poolTokenAmount`) for the equivalent amount of WETH. However, the function incorrectly calls `swapExactOutput` instead of `swapExactInput`.

Since users specify the exact number of pool tokens they want to sell (input), the swap should be processed using `swapExactInput`. By using `swapExactOutput`, the logic instead assumes the user is targeting a desired output amount, which misaligns the calculation and results in the wrong amount of tokens being transferred.

Impact:

- Users may receive an incorrect amount of WETH for their pool tokens.
- This breaks the expected behavior of the `sellPoolTokens` function and can cause severe financial loss or complete malfunction of a core feature in the protocol.

- If integrated into a UI, users will consistently experience failed or incorrect swaps, undermining trust in the protocol.

Proof of Concept: A user calls `sellPoolTokens(1e18)`, expecting to receive the equivalent WETH for their pool tokens. Instead, since the function uses `swapExactOutput`, the protocol interprets the input as a desired output, leading to either zero tokens received or an amount that does not match the pool token input.

Recommended Mitigation: Update the `sellPoolTokens` implementation to use `swapExactInput` instead of `swapExactOutput`, since users specify the exact number of pool tokens they want to sell. Additionally, introduce a `minWethToReceive` parameter to provide slippage protection, ensuring users are guaranteed a minimum amount of WETH.

```
1     function sellPoolTokens(  
2         uint256 poolTokenAmount,  
3 +         uint256 minWethToReceive  
4     ) external returns (uint256 wethAmount) {  
5 -         return swapExactOutput(i_poolToken, i_wethToken,  
6 +         poolTokenAmount, uint64(block.timestamp));  
7         return swapExactInput(i_poolToken, poolTokenAmount, i_wethToken  
            , minWethToReceive, uint64(block.timestamp));  
            }
```

Additionally, it might be wise to add a deadline to the function, as there is currently no deadline.

Medium Severity

[M-1] TSwapPool::deposit does not enforce deadline, risking post-deadline execution

Description: The `deposit` function accepts a deadline parameter, which according to the documentation is “The deadline for the transaction to be completed by”. However, this parameter is never used. As a consequence, operations that add liquidity to the pool might be executed at unexpected times, in market conditions where the deposit rate is unfavorable.

Impact: Transaction could be sent when market conditions are unfavorable to deposit, even when adding a deadline parameter.

Proof of Concept: The `deadline` parameter is unused.

Recommended Mitigation: Consider making the following change to the function.

```
1     function deposit(  
2         uint256 wethToDeposit,  
3         uint256 minimumLiquidityTokensToMint,  
4         uint256 maximumPoolTokensToDeposit,
```

```
5         uint64 deadline
6     )
7     external
8 +     revertIfDeadlinePassed(deadline)
9     revertIfZero(wethToDeposit)
10    returns (uint256 liquidityTokensToMint)
11    {...}
```

Low Severity

[L-1] Liquidity token symbol uses `.name()` instead of `.symbol()` in `PoolFactory::createPool`

Description: The liquidity token symbol is generated using the token's name instead of its symbol. This can cause confusion in user interfaces, dashboards, and third-party integrations (such as DEXs) that rely on the correct token symbol for identification. Users may see misleading or inconsistent information, potentially leading to mistakes when interacting with the pool.

```
1     function createPool(address tokenAddress) external returns (address
2         ) {
3         if (s_pools[tokenAddress] != address(0)) {
4             revert PoolFactory__PoolAlreadyExists(tokenAddress);
5         }
6         string memory liquidityTokenName = string.concat("T-Swap ",
7             IERC20(tokenAddress).name());
8 -->    string memory liquidityTokenSymbol = string.concat("ts", IERC20
9         (tokenAddress).name());
10
11        TSwapPool tPool = new TSwapPool(tokenAddress, i_wethToken,
12            liquidityTokenName, liquidityTokenSymbol);
13        s_pools[tokenAddress] = address(tPool);
14        s_tokens[address(tPool)] = tokenAddress;
15        emit PoolCreated(tokenAddress, address(tPool));
16        return address(tPool);
17    }
```

Impact: Using `.name()` instead of `.symbol()` can confuse users and third-party interfaces, showing misleading token symbols without affecting funds.

Proof of Concept:

1. Deploy a pool using `PoolFactory::createPool` for any ERC20 token.
2. Observe that the liquidity token symbol is incorrectly derived from the token's name:

```
string memory liquidityTokenSymbol = string.concat("ts", IERC20(
tokenAddress).name());
```

3. On wallets, UIs, or DEXs, the liquidity token displays a misleading symbol, e.g., "T-Swap" instead of "ts".
4. Integrations relying on the token symbol may display or categorize it incorrectly.

Recommended Mitigation: Replace the use of `.name()` with `.symbol()` when generating the liquidity token symbol, as shown in the code snippet.

```
1 string memory liquidityTokenName = string.concat("T-Swap ", IERC20(
tokenAddress).name());
2 - string memory liquidityTokenSymbol = string.concat("ts", IERC20(
tokenAddress).name());
3 + string memory liquidityTokenSymbol = string.concat("ts", IERC20(
tokenAddress).symbol());
```

[L-2] TSwapPool::_addLiquidity event has parameters out of order

Description: When the `LiquidityAdded` event is emitted in the `TSwapPool::_addLiquidityMintAndTrans` function, it logs values in an incorrect order. The `poolTokensToDeposit` value should go in the third parameter position, where as the `wethToDeposit` value should go second.

Impact: Event emission is incorrect, leading to off-chain functions potentially malfunctioning.

Proof of Concept:

Recommended Mitigation:

```
1 - emit LiquidityAdded(msg.sender, poolTokensToDeposit, wethToDeposit)
;
2 + emit LiquidityAdded(msg.sender, wethToDeposit, poolTokensToDeposit)
;
```

[L-3] TSwapPool::swapExactInput returns default zero due to uninitialized output variable

Description: The `swapExactInput` function is expected to return the actual amount of tokens bought by the caller. However, the declared return variable `output` is never assigned a value nor explicitly returned. As a result, the function always returns 0, providing incorrect information to callers and misleading external integrations.

Impact: Callers and external contracts relying on the return value receive 0 instead of the actual amount swapped, which can cause incorrect accounting, UI misrepresentation, or unexpected behavior in automated strategies.

Proof of Concept: The following test demonstrates this behavior in practice.

Proof of Code:

Code

```
1      function test_swapExactInput_returnsDefaultZero() public {
2          uint256 initialLiquidity = 10e18;
3          uint256 swapAmount = 4e18;
4
5          // --- Arrange: provide initial liquidity to the pool ---
6          vm.startPrank(liquidityProvider);
7          weth.approve(address(pool), initialLiquidity);
8          poolToken.approve(address(pool), initialLiquidity);
9          pool.deposit(initialLiquidity, initialLiquidity,
10                      initialLiquidity, uint64(block.timestamp));
11          vm.stopPrank();
12
13          // --- Act: user attempts to swap tokens ---
14          vm.startPrank(user);
15          weth.approve(address(pool), swapAmount);
16          poolToken.approve(address(pool), swapAmount);
17          uint256 outputAmount = pool.swapExactInput(weth, swapAmount,
18              poolToken, 0, uint64(block.timestamp));
19          vm.stopPrank();
20
21          // --- Assert: the returned output is zero due to missing
22          //      assignment ---
23          assertEq(outputAmount, 0, "swapExactInput should return 0 when
24              output is not assigned");
25      }
```

Recommended Mitigation: Assign the calculated output to the return variable to ensure the function returns the correct swapped amount. This prevents misleading results and potential issues in UI and integrations.

```
1      {
2          uint256 inputReserves = inputToken.balanceOf(address(this));
3          uint256 outputReserves = outputToken.balanceOf(address(this));
4
5          -      uint256 outputAmount = getOutputAmountBasedOnInput(inputAmount,
6              inputReserves, outputReserves);
7          +      output = getOutputAmountBasedOnInput(inputAmount, inputReserves
8              , outputReserves);
9
10         -      if (outputAmount < minOutputAmount) {
11         -          revert TSwapPool__OutputTooLow(outputAmount,
12             minOutputAmount);
13         -      }
14         +      if (output < minOutputAmount) {
15         +          revert TSwapPool__OutputTooLow(output, minOutputAmount);
16         +      }
```

```
13 +     }
14
15 -     _swap(inputToken, inputAmount, outputToken, outputAmount);
16 +     _swap(inputToken, inputAmount, outputToken, output);
17 }
```

Informational

[I-1] PoolFactory::PoolFactory__PoolDoesNotExist is unused

Description: The error `PoolFactory__PoolDoesNotExist` is declared but never used. Keeping unused errors adds unnecessary complexity and may confuse developers.

Recommended Mitigation:

```
1 - error PoolFactory__PoolDoesNotExist(address tokenAddress);
```

[I-2] Missing zero-address validation in PoolFactory constructor

Description: Constructor does not validate `wethToken` against `address(0)`, allowing deployment with an invalid WETH address.

Impact:

- PoolFactory may deploy pools referencing `address(0)`, causing token calls to revert or behave incorrectly.
- Breaks integrations that assume a valid WETH token; potential DoS and costly migration.

Proof of Concept:

1. Deploy `PoolFactory` with `wethToken == address(0)`.
2. Attempt to create or interact with pools — token calls will fail or produce incorrect bookkeeping.

Recommended Mitigation: Validate input in the constructor and provide a clear error message.

```
1     constructor(address wethToken) {
2 +         if (wethToken == address(0)) {
3 +             revert();
4 +         }
5         i_wethToken = wethToken;
6     }
```

[I-3] Missing zero-address validation in TSwapPool constructor

Description: Constructor does not validate `i_wethToken` and `i_poolToken` against `address(0)`, allowing deployment with invalid token addresses.

Impact:

- Pool functions will fail or behave incorrectly (DoS, broken accounting).
- Unexpected state / invariant violations and costly migration to fix.

Proof of Concept:

1. Deploy `TSwapPool` with `i_wethToken == address(0)` (or `i_poolToken == address(0)`).
2. Calls that interact with the token (e.g., `deposit`, `swap`) will revert or not transfer funds, breaking pool logic.

Recommended Mitigation: Add zero-address checks in the constructor and unit tests.

```
1     constructor(  
2         address poolToken,  
3         address wethToken,  
4         string memory liquidityTokenName,  
5         string memory liquidityTokenSymbol  
6     )  
7     ERC20(liquidityTokenName, liquidityTokenSymbol)  
8     {  
9 +     if (wethToken == address(0) || poolToken == address(0)) {  
10 +         revert();  
11 +     }  
12     i_wethToken = IERC20(wethToken);  
13     i_poolToken = IERC20(poolToken);  
14 }
```

[I-4] Event TSwapPool : : Swap has more than 3 parameters and should index key fields

Description: The `Swap` event has five parameters but only the `swapper` is indexed. Indexing additional key parameters (`tokenIn` and `tokenOut`) improves filtering and efficiency for off-chain listeners.

Recommended Mitigation:

```
1 - event Swap(address indexed swapper, IERC20 tokenIn, uint256  
   amountTokenIn, IERC20 tokenOut, uint256 amountTokenOut);  
2 + event Swap(address indexed swapper, IERC20 indexed tokenIn, uint256  
   amountTokenIn, IERC20 indexed tokenOut, uint256 amountTokenOut);
```


[I-5] Missing `deadline` NatSpec in `TSwapPool::swapExactOutput`

Description: `swapExactOutput` accepts a `deadline` parameter but the NatSpec is missing.

```
1      /*
2      * @notice figures out how much you need to input based on how much
3      * output you want to receive.
4      *
5      * Example: You say "I want 10 output WETH, and my input is DAI"
6      * The function will figure out how much DAI you need to input to
7      * get 10 WETH
8      * And then execute the swap
9      * @param inputToken ERC20 token to pull from caller
10     * @param outputToken ERC20 token to send to caller
11     * @param outputAmount The exact amount of tokens to send to caller
12     --> */
13     function swapExactOutput(
```

Recommended Mitigation: Add a concise `@param` entry describing the parameter and expected behavior.

[I-6] CEI pattern not followed in `TSwapPool::deposit`

Description: In the initial deposit branch, `liquidityTokensToMint` is assigned after calling `_addLiquidityMintAndTransfer`. Following CEI consistently—even for local variables—improves clarity and reduces potential mistakes.

Recommended Mitigation:

```
1 -         _addLiquidityMintAndTransfer(wethToDeposit,
2 -             maximumPoolTokensToDeposit, wethToDeposit);
3 -         liquidityTokensToMint = wethToDeposit;
4 +         liquidityTokensToMint = wethToDeposit;
5 +         _addLiquidityMintAndTransfer(wethToDeposit,
6 +             maximumPoolTokensToDeposit, wethToDeposit);
```

[I-7] Missing NatSpec in `TSwapPool` functions

Description: `getOutputAmountBasedOnInput`, `getInputAmountBasedOnOutput`, and `swapExactInput` lack NatSpec documentation. Without it, users and integrators cannot easily understand function purpose, inputs, and outputs. Providing NatSpec improves readability, usability, and auditing.

Recommended Mitigation: Add NatSpec for each function.

Gas Optimization

[G-1] `TSwapPool::swapExactInput` can be `external`

Description: `swapExactInput` is not called internally; marking it `external` saves gas and signals intended usage.

Recommended Mitigation:

```
1     function swapExactInput(  
2         ...  
3     )  
4 -     public  
5 +     external  
6         revertIfZero(inputAmount)  
7         revertIfDeadlinePassed(deadline)  
8         returns (uint256 output)  
9     {  
10        ...  
11    }
```